

## **Excel Tips: Text it Up** (altavia.com, 2016)

I find Excel text functions extremely helpful in manipulating presented data into a format I can use. For example, cost centers are often numerical characters of a specified length with preceding 0's. So, if you are faced with a column of cost centers that are formatted as numbers, and you need to make them all 3 characters long with preceding 0's, you would use the RIGHT and CONCATENATE commands to make quick work of this issue.

### **RIGHT Function**

The RIGHT function syntax is: RIGHT(text, num\_chars)

The concatenate "&" function syntax is: string1 & string2

Assuming the first incorrectly formatted cost center is "5" in cell A2, this is what the formula looks like:

=RIGHT("00" & A2, 3) That's it!

What this says: take two zero's, "00", put that in front of whatever is in column A2, then take the three characters starting from the right, resulting in the text field "005." Notice in row 6 below, there are no preceding zero's as the command is only reading 3 characters from the right.

	A	B	C
1	<b>Cost Ctr</b>	<b>Formula</b>	<b>Result</b>
2	5	=RIGHT("00"&A2,3)	005
3	9	=RIGHT("00"&A3,3)	009
4	12	=RIGHT("00"&A4,3)	012
5	35	=RIGHT("00"&A5,3)	035
6	123	=RIGHT("00"&A6,3)	123

### **LEFT Function**

The LEFT function syntax is: LEFT(text, num\_chars)

Here's an example of using the LEFT command:

	A	B	C
1	<b>Part # - Source</b>	<b>Formula</b>	<b>Part #</b>
2	4598-Azt59K	=LEFT(A2,4)	4598
3	2368-Qrs38I	=LEFT(A3,4)	2368
4	6987-Zgy12L	=LEFT(A4,4)	6987
5	2596-Plx36Y	=LEFT(A5,4)	2596
6	3698-Yjh78k	=LEFT(A6,4)	3698

### **FIND Function**

Now, let's say things aren't so structured as in the above example, and the part numbers are different lengths, followed by a dash and the Source. In this case we will use the FIND command and the dash to tell us how many characters from the left to go.

The FIND function syntax is: FIND(find\_text, within\_text)

Note that the FIND command returns the position number of the character it finds.

Let's say a random length part number is 12359875, found in string 12359875-Azt59K, is in cell A2. This is what the formula looks like:

=LEFT(A2, FIND("-",A2) -1)

What this says: find the dash "-" in the string in cell A2, which is at position number 9 in our example, subtract 1 to yield 8, then take the first 8 characters from the left of the string, resulting in the text field "12359875."

	A	B	C
1	<b>Part # - Source&amp;Yr</b>	<b>Formula</b>	<b>Part #</b>
2	12359875-Azt59K2015	=LEFT(A2,FIND("-",A2)-1)	12359875
3	125-Qrs38I2015	=LEFT(A3,FIND("-",A3)-1)	125
4	235689-Zgy12L2015	=LEFT(A4,FIND("-",A4)-1)	235689
5	25687-Plx36Y2015	=LEFT(A5,FIND("-",A5)-1)	25687
6	201512569871233-Yjh78K2015	=LEFT(A6,FIND("-",A6)-1)	201512569871233

### **Mid Function**

A related function is the MID function, which will return the text starting in the middle of another text string and go a defined number of characters.

The MID function syntax is: MID(text, start\_num, num\_chars)

If we wanted to get the 6 character Source rather than the Part Number or Year in the above example, the command would be:

=MID(A2,FIND("-",A2)+1,6)

What this says: find the dash "-" in the string in cell A2, which is position number 9 in our example, add 1 to yield 10, so start at character position 10 in the text string in cell A2 and return the next 6 characters, yielding the result "Azt59K."

	A	B	C
1	<b>Part # - Source</b>	<b>Formula</b>	<b>Source</b>
2	12359875-Azt59K2015	=MID(A2,FIND("-",A2)+1,6)	Azt59K
3	125-Qrs38I2015	=MID(A3,FIND("-",A3)+1,6)	Qrs38I
4	235689-Zgy12L2015	=MID(A4,FIND("-",A4)+1,6)	Zgy12L
5	25687-Plx36Y2015	=MID(A5,FIND("-",A5)+1,6)	Plx36Y
6	201512569871233-Yjh78K2015	=MID(A6,FIND("-",A6)+1,6)	Yjh78K

Another way to solve the same problem by combining commands:

	A	B	C
1	<b>Part # - Source</b>	<b>Formula</b>	<b>Source</b>
2	12359875-Azt59K2015	=LEFT(RIGHT(A2,10),6)	Azt59K
3	125-Qrs38I2015	=LEFT(RIGHT(A3,10),6)	Qrs38I
4	235689-Zgy12L2015	=LEFT(RIGHT(A4,10),6)	Zgy12L
5	25687-Plx36Y2015	=LEFT(RIGHT(A5,10),6)	Plx36Y
6	201512569871233-Yjh78K2015	=LEFT(RIGHT(A6,10),6)	Yjh78K

### **A Common Example**

Now let's use these commands to break an address into its components:

	A	B	C
1	<b>Address</b>	<b>Formula</b>	<b>Street, City, State, Zip</b>
2	12345 Apple Way, Arlington, VA 22205	=LEFT(A2,FIND(",",A2)-1)	12345 Apple Way
3		=MID(A2,FIND(",",A2)+2,99)	Arlington, VA 22205
4		=LEFT(C3,FIND(",",C3)-1)	Arlington
5		=MID(C3,FIND(",",C3)+2,2)	VA
6		=RIGHT(C3,5)	22205

Note that the result of the formula in cell B3 is used by the formulas in cells B4:B6.

### **Other Text Functions**

Finally for this post, here are some simple yet extremely helpful text functions.

LEN(text)      Number of characters in the text string (text can refer to a cell, such as A2)  
 TRIM(text)      Trim off preceding or following spaces, if any (use when comparing 2 text strings)  
 UPPER(text)      Change all characters to Upper Case  
 LOWER(text)      Change all characters to Lower Case  
 PROPER(text)      Change first character of each word to Upper Case, all other characters Lower Case

Formula on, dudes...